
ABSTRACT

What if the usual sort and usual group processing would eliminate the existing logical order and grouping of the data that must be maintained to ensure the needed analysis is correct.  This paper presents some uses of the NOTSORTED option of the BY statement when working with group processing in data steps and procedure steps.  As an extra, the appendix describes the use of the keyword NOTSORTED with respect to views, indexes, and user-defined formats. 

INTRODUCTION

Beginning SAS® programmers soon realize that the proc sort procedure is one of the most frequently used SAS procedures.  Often, proc sort with a BY statement is used at the start of some type of group processing.  But what if the usual sort and the usual group processing would eliminate the desired logical order and logical grouping of the data and consequently produce an inappropriate analysis.   This paper presents to intermediate SAS programmers some uses of the NOTSORTED option of the BY statement when working with group processing in data steps and procedure steps.

This paper first describes a data problem and how it can arise.  It then addresses the problem by presenting three code solutions.  All three solutions are based on the power of the NOTSORTED option.  Then we examine how the NOTSORTED option in the BY statement impacts data step processing.  Next we examine how the NOTSORTED option in the BY statement impacts procedure processing and the processing of various commonly used procedures.  As an extra, the appendix of this paper describes the use of the keyword NOTSORTED with respect to views, indexes, and user-defined formats.  The actual outputs may be simplified for display purposes.  To simplify the discussion and display, we use small illustrative data sets, such as DS1 (below).

A DATA ANALYSIS PROBLEM WITH SOLUTIONS USING BY’S NOTSORTED OPTION

In this section, a data analysis problem is described and analyzed.  Then three solutions are presented.

· A data step solution

· A procedure step solution

· Pre-processing solution

Each solution is based on the power of the NOTSORTED option.

A Data Analysis Problem

Throughout this section and much of this paper, we use the data set: DS1.

DATA SET: DS1       STATE        AMT    
                    PA           200

                    PA           100

                    CA           300

                    CA           200

                    MA           200

                    MA           400

                    PA           100

                    PA           300

The following typical code creates a typical summary data set SUMDS1.

CODE    proc sort data=DS1 out=DS1s;

               by STATE; run;

        data SUMDS1 (drop=AMT);

        set  DS1s;

        by STATE;

        if first.STATE then TOTAMT = 0;

        TOTAMT+AMT;

        if last.STATE then output; run;

DATA SET: SUMDS1    STATE        TOTAMT

                    CA           500

                    MA           600

                    PA           700

Notice that the original data set DS1 is grouped but the groups are not ordered in either ascending or descending order and that there are actually two groupings for the STATE variable with value ‘PA’.   The problem appears because the client needs an analysis where the summary data set must look like SUMDS2.  

DATA SET: SUMDS2    STATE        TOTAMT

                    PA           300

                    CA           500

                    MA           600

                    PA           400

With the sort we get both an inappropriate order and grouping.  If the sort is not executed and the above data step is executed for DS1 the SAS Log contains the error message:

OUTPUT ERROR: BY variables are not properly sorted on data set WORK.DS1.

Also by the relational nature of SQL, proc sql cannot do BY-NOTSORTED group processing.  Consequently, for SAS beginners, there is no self-evident way to manipulate DS1 and output SUMDS2.  

A Data Step Solution

As usual, we can rely on base SAS data step coding for one or more solutions.  The above code can easily be modified to provide a short and natural solution by removing the proc sort call and by making only a few changes to the above data step code.  The solution uses the power of the BY statement with the NOTSORTED option.

CODE    data SUMDS2 (drop=AMT);

        set  DS1;

        by STATE NOTSORTED;
        if first.STATE then TOTAMT = 0;

        TOTAMT+AMT;

        if last.STATE then output; run;

A Procedure Solution

This solution also uses the power of the BY statement with the NOTSORTED option.  Using only a CLASS statement will not give the desired data set SUMDS2.  This code provides a short and natural solution and does not use proc sort pre-processing.

CODE    proc means data=DS1 sum maxdec=0  noprint;

               by STATE NOTSORTED;
        var AMT; output out=sumds2(drop=_type_ _freq_ ) sum(amt)=TOTAMT ; run;

A Pre-processing Solution

This solution also uses the power of the BY statement with the NOTSORTED option.  The following code adds a  ‘grouping’ column GrpNum such that any followup analysis can use the GrpNum column for sorting and grouping.  Using the grouping column in this way, we can avoid the further use of the NOTSORTED option in any follow-up analysis. 

CODE    data DS2;

        set  DS1;

        retain grpnum 0;

        by STATE NOTSORTED;
        if first.STATE then grpnum=grpnum+1;

        else grpnum=grpnum;

        output; run;

DATA SET: DS2       STATE        AMT        GRPNUM    
                    PA           200        1

                    PA           100        1

                    CA           300        2

                    CA           200        2

                    MA           200        3

                    MA           400        3

                    PA           100        4

                    PA           300        4

CODE    data SUMDS2 (drop=AMT GRPNUM);

        set  DS2;

        by GRPNUM;  /* NO  USE OF  NOTSORTED */

        if first.GRPNUM then TOTAMT = 0;

        TOTAMT+AMT;

        if last.GRPNUM then output; run;

THE DATA STEP’S BY STATEMENT WITH THE NOTSORTED OPTION 

Variables in a BY statement are called BY variables.  A BY group is a set of contiguous rows (observations) that have the same values for all BY variables.   The BY variables are used to form BY groups.

The BY statement applies only to the SET, MERGE, MODIFY, or UPDATE statement that immediately precedes it in the DATA step, and only one BY statement can accompany each of these statements in a data step.  

Now we examine how the NOTSORTED option in the BY statement impacts data step processing.  If you do not use the NOTSORTED option in the BY statement, the rows in the data set must either be sorted by the BY variables specified or the data set must be indexed appropriately.  That is, NOTSORTED specifies that the rows are not necessarily in sorted order, but are logically grouped in some other way (perhaps chronological order or in categories).  In other words, the NOTSORTED option does not mean the data is unsorted, rather that the data is arranged in groups (according to the values of the BY variables) and that the groups are not necessarily in ascending or descending order.  The NOTSORTED option informs SAS that the rows are grouped by the BY variables, but are not presented in a sorted order.  Briefly, NOTSORTED indicates that BY group processing takes place on grouped, rather than sorted data.  Anytime any one of the BY variables changes value, SAS is to interpret this as a new BY group.  Once again, if observations with the same values for the BY variables are not contiguous, a data step with BY-NOTSORTED processing treats each contiguous set as a separate BY group.  

The data sets that are listed in the SET, MERGE, or UPDATE statements must be sorted by the values of the variables that are listed in the BY statement or have an appropriate index.  As a default, SAS expects the data sets to be in ascending order.  Consequently, NOTSORTED cannot be used with the MERGE statement, UPDATE statement, or when the SET statement lists more than one data set.  For completeness, MODIFY does not require sorted data, but sorting can improve performance.   When using the SET statement with multiple data sets (interleaving) and the NOTSORTED option is specified in the BY statement, unpredictable row groupings could result and the following error message will be produced: 

OUTPUT  BY NOTSORTED MAY NOT BE USED WITH SET STATEMENT WHEN MORE THAN ONE DATA SET IS

        SPECIFIED.

The NOTSORTED option can appear once or more times anywhere in the BY statement.  For example code, see a previous section, ‘A Data Analysis Problem With Solutions Using BY’s NOTSORTED Option’.

The BY statement also has the GROUPFORMAT option.  This option specifies the data step is to use the formatted values, instead of the internal values, of the BY variables to determine where BY groups begin and end, and consequently how FIRST.var and LAST.var get assigned.  Although the GROUPFORMAT option can appear anywhere in the BY statement, the option applies to all BY variables.  If, in addition, the NOTSORTED option is used the rows can be grouped by the formatted value of the BY variables without needing the data set to be sorted or indexed.

CODE    proc format; 

        value $stfmt 'MA'='Small'  'PA','CA'='Large'; run;

        data state; 

        set ds1; 

        by state GROUPFORMAT NOTSORTED; 

        format state $stfmt.; run;

        proc print data=state;

        by  state NOTSORTED;

        sum amt;  run;

OUTPUT  The SAS System   16:59 Monday, December 26, 2005   1

        --STATE=Large -------

        Obs      AMT

        1        200

        2        100

        3        300

        4        200

        -----    ---

        STATE    800

        --STATE=Small --------

        Obs      AMT

        5        200

        6        400

        -----    ---

        STATE    600

        --STATE=Large ---------

        Obs      AMT

        7        100

        8        300

        -----    ----

        STATE    400

                ====

                1800
THE PROCEDURE STEP’S BY STATEMENT WITH THE NOTSORTED OPTION 
Recall, variables in a BY statement are called BY variables.  A BY group is a set of contiguous rows (observations) that have the same values for all BY variables.  The BY variables are used to form BY groups. 

Now we examine how the NOTSORTED option in the BY statement impacts procedure processing for various commonly used procedures.  First we make comments that apply to procedures that use the BY statement and that allow the NOTSORTED option on the BY statement.   Next, in the many subsections, we examine how the NOTSORTED option in the BY statement impacts several frequently used types of procedure steps.  As the usage of the BY statement differs in each procedure, refer to the SAS documentation for details.

A procedure does not use an index if you specify NOTSORTED.  More accurately, when you use the NOTSORTED option the requirement for ordering or indexing rows according to the values of BY variables is suspended for BY-group processing.  

Only one BY statement can be used in each PROC step.  A procedure creates output for each BY group.  If observations with the same values for the BY variables are not contiguous, a procedure with BY-NOTSORTED processing treats each contiguous set as a separate BY group.   The statistics procedures perform separate analyses for each BY group and the reporting procedures produce a report for each BY group.

A procedure with a BY statement expects an input data set that is sorted by the order of the BY variables or one that has an appropriate index.  An error occurs if the input data set does not meet these criteria.  Either sort it with the PROC SORT or create an appropriate index on the BY variables or use the NOTSORTED or DESCENDING option in the procedure step BY statement.

Here are a few representative procedures supporting the BY statement with the NOTSORTED option:  btl, calendar, chart, compare, corr, forms, freq, glm, means, mi, nested, plot, print, rank, report, score, standard, summary, tabulate, timeplot, transpose, and univariate.  Most of the general comments about the BY statement NOTSORTED option for a procedure step apply to the various SAS statistical procedure steps, such as btl, corr, glm, nested, score, and standard.  The following subsections make additional comments and provide examples for several of the more common procedure steps and how the NOTSORTED option on the BY statement impacts procedure processing.

The Sort Procedure Step’s By Statement With The NOTSORTED Option
The BY statement in the SORT procedure specifies how to sort the data.  Consequently, the NOTSORTED option cannot be used in a PROC SORT step.  In other procedures, the BY statement specifies how the data are currently sorted.

When a data set is sorted using the EBCDIC collating sequence, you must use the NOTSORTED option on any subsequent BY statements that refer to the data set.  Without the NOTSORTED option, you will receive a message that the data set is not in sorted order.  

DATA SET: DSeb1  STATE        AMT

                 CA           300

                 CA           200

                 MA           200

                 MA           400

                 pa           100

                 pa           300

CODE    proc sort data=DSeb1 out=DSeb1s EBCDIC;

        by state; run;
DATA SET: DSeb1s       STATE        AMT

                       pa           100

                       pa           300

                       CA           300

                       CA           200

                       MA           200

                       MA           400

The following produces an error message:

CODE    proc print data=DSeb1s; 

        by state; run;

OUTPUT  ERROR: Data set WORK.DS8EBCDIC is not sorted in ascending sequence. The current

        by-group has STATE = pa and the next by-group has STATE = CA.

The following produces what is needed

CODE    proc print data=DSeb1s; 

        by state NOTSORTED; run;

OUTPUT  The SAS System        20:42 Monday, December 26, 2005   1

        ---- STATE=pa --------

        Obs    AMT

        1     100

        2     300

        ---- STATE=CA --------

        Obs    AMT

        3     300

        4     200

        ---- STATE=MA --------

        Obs    AMT

        5     200

        6     400

The SORTEDBY= data set option can specify how a data set is currently sorted.  If a non-native collating sequence or EBCIDIC is specified in the SORTEDBY= option, subsequent group processing on the data set requires the NOTSORTED option on the BY statement.  Without the use of NOTSORTED on the BY statement, you will receive a message that the variables are not properly sorted.  

The Print Procedure Step’s BY Statement With The NOTSORTED Option
All base procedures except PROC PRINT process BY groups completely independently.  PROC PRINT is able to report the number of rows in each BY group as well as the number of rows in all BY groups.  Also, PROC PRINT can sum numeric variables in each BY group and across all BY groups.

We will see the results of a) not using the NOTSORTED option, b) positioning NOTSORTED at various locations in the BY statement, c) using NOTSORTED multiple times in the BY statement, d) and not specifying any by variables in the BY-NOTSORTED statement. 

DATA SET: STCNTY    ST      COUNTY      AMT    
                    PA      03          300

                    CA      01          500

                    CA      01          100

                    MA      01          600

                    PA      03          600

                    PA      02          200

The following produces an error message:

CODE    proc print data=stcnty;

        by st county; run;

OUTPUT  ERROR: Data set WORK.STCNTY is not sorted in ascending sequence. 

        The current by-group has st = PA and the next by-group has st = CA.

        NOTE: The SAS System stopped processing this step because of errors.

The following produces what is needed:

CODE    proc print data=stcnty;

        by NOTSORTED  st  county; 

        sum amt;  run;

OUTPUT  The SAS System       20:42 Monday, December 26, 2005   1

        ---- ST=PA COUNTY=03 --

         Obs      AMT

           1      300

        ---- ST=CA COUNTY=01 --

         Obs      AMT

           2      500

           3      100

        ------    ---

        COUNTY    600

        ST        600

        ---- ST=MA COUNTY=01 ---

         Obs      AMT

           4      600

        ---- ST=PA COUNTY=03 ---

         Obs      AMT

           5      600

        ---- ST=PA COUNTY=02 ---

         Obs      AMT

           6      200

        ------    ----

        ST        800

                  ====

                 2300

CODE    proc print data=stcnty;

        by NOTSORTED  st  NOTSORTED  county ; run;

OUTPUT  The same as the last output.

The following code produces similar results with a different layout:

CODE    proc print data=stcnty;

        by NOTSORTED; sum  amt;  run;

OUTPUT  The SAS System    20:42 Monday, December 26, 2005   1
        Obs   st    county    amt

        1     PA      03      300

        2     CA      01      500

        3     CA      01      100

        4     MA      01      600

        5     PA      03      600

        6     PA      02      200

                              ===

                             2300

The Means Procedure Step’s BY Statement With The NOTSORTED Option
Proc means runs significantly faster for large numbers of groups if the data is first sorted with proc sort and then a BY statement with the classification variables instead of a CLASS statement.  The CLASS statement informs the means procedure to build a table of all possible combinations of class variables.  The BY statement allows proc means to process a single group, then write it to the output data set, and access the same storage to process the next group.  The BY statement needs either a sorted data set or a grouped data set with the NOTSORTED option on the BY statement.  

For example code, see a previous section, ’A Data Analysis Problem With Solutions Using BY’s NOTSORTED Option’.

DATA SET: DSmean1   STATE        AMT    
                    PA           200

                    PA           100

                    CA           300

                    MA           400

                    PA           300

CODE    proc means data=DSmean1 mean maxdec=0; 

               by state NOTSORTED;

               class state  / order=data;

               var amt; run;

OUTPUT  The SAS System    20:42 Monday, December 26, 2005   1

               STATE=PA --------------------

               The MEANS Procedure

               Analysis Variable: AMT

               STATE    N Obs          Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

               PA         2             150

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

               STATE=CA -------------------

               The MEANS Procedure

               Analysis Variable: AMT

               STATE    N Obs          Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

               CA         1             300

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

               STATE=MA -------------------

               The MEANS Procedure

               Analysis Variable: AMT

               STATE    N Obs          Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

               MA         1             400

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

               STATE=PA -------------------

               The MEANS Procedure

               Analysis Variable: AMT

               STATE    N Obs          Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

               PA         1             300

               ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

The following does not use a CLASS statement.

CODE    proc means data=DSmean1 mean maxdec=0;

               by state NOTSORTED;

               var amt; run;

OUTPUT  The SAS System    20:42 Monday, December 26, 2005   1

               STATE=PA --------------

               The MEANS Procedure

               Analysis Variable: AMT

               Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒ

               150

               ƒƒƒƒƒƒƒƒƒƒƒƒ

               STATE=CA --------------

               Analysis Variable: AMT

               Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒ

               300

               ƒƒƒƒƒƒƒƒƒƒƒƒ

               STATE=MA --------------

               Analysis Variable: AMT

               Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒ

               400

               ƒƒƒƒƒƒƒƒƒƒƒƒ

               STATE=PA --------------

               Analysis Variable: AMT

               Mean

               ƒƒƒƒƒƒƒƒƒƒƒƒ

               300

               ƒƒƒƒƒƒƒƒƒƒƒƒ

The Univariate Procedure Step’s BY Statement With The NOTSORTED Option
When categorical variables are created from a continuous variable, proc univariate can be used to check the accuracy, with code like:

CODE    proc univariate data=ds1 noprint;

        by state NOTSORTED; 

        output out=sumry  n=NonMissing median=Median qrange=QRange; run;

DATA SET: SUMRY     STATE   NonMissing   Median    QRange    
                    PA      2            300       100

                    CA      2            500       100

                    MA      2            100       200

                    PA      2            600       200

The Freq Procedure Step’s BY Statement With The NOTSORTED Option
Proc freq shows the distribution of categorical data values and can reveal some data irregularities.  The first sample code shows the usual proc freq analysis for  data set DS1.

Notice the output has grouped all the PA data into a single group.  However, the second sample code that uses the power of the NOTSORTED option produces a separate analysis for each by grouping.  Specifically, there are two by groupings for State=‘PA’.  

CODE    proc freq data=ds1;

        table state;

        run;

OUTPUT  The SAS System         20:42 Monday, December 26, 2005   1
        The FREQ Procedure

                                          Cumulative    Cumulative

        STATE    Frequency     Percent     Frequency      Percent

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        CA              2       25.00             2        25.00

        MA              2       25.00             4        50.00

        PA              4       50.00             8       100.00

The following code uses a BY-NOTSORTED statement:  

CODE    proc freq data=ds1;

        by  state NOTSORTED;

        table state  / nopercent nocum;  run;

OUTPUT  The SAS System         20:42 Monday, December 26, 2005   1
        ------------------------- STATE=PA -----------------------

        The FREQ Procedure

        STATE    Frequency

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        PA              2

        The SAS System         20:42 Monday, December 26, 2005   2
        ------------------------ STATE=CA ------------------------

        The FREQ Procedure

        STATE    Frequency

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        CA              2 

        The SAS System         20:42 Monday, December 26, 2005   3
        ------------------------- STATE=MA -----------------------

        The FREQ Procedure

        STATE    Frequency

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        MA              2

        The SAS System         20:42 Monday, December 26, 2005   4
        ------------------------- STATE=PA -----------------------

        The FREQ Procedure

        STATE    Frequency

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        PA              2 

The Tabulate Procedure Step’s BY Statement With The NOTSORTED Option
Proc tabulate can produce much of what is done by proc print, proc means, and proc freq.  However, the reports produced by proc tabulate are prettier.
CODE    proc tabulate data=ds1;

        by state NOTSORTED;

        class state; 

        table state;  run;
OUTPUT  The SAS System     20:42 Monday, December 26, 2005   1

        --- STATE=PA -----------------------------------------

        „ƒƒƒƒƒƒƒƒƒƒƒƒ†

        ‚   STATE    ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     PA     ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     N      ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚        2.00‚

        ŠƒƒƒƒƒƒƒƒƒƒƒƒŒ

        The SAS System     20:42 Monday, December 26, 2005   2

        --- STATE=CA -----------------------------------------

        „ƒƒƒƒƒƒƒƒƒƒƒƒ†

        ‚   STATE    ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     CA     ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     N      ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚        2.00‚

        ŠƒƒƒƒƒƒƒƒƒƒƒƒŒ

        The SAS System     20:42 Monday, December 26, 2005   3

        --- STATE=MA -----------------------------------------

        „ƒƒƒƒƒƒƒƒƒƒƒƒ†

        ‚   STATE    ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     MA     ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     N      ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚        2.00‚

        ŠƒƒƒƒƒƒƒƒƒƒƒƒŒ

        The SAS System     20:42 Monday, December 26, 2005   4

        --- STATE=PA -----------------------------------------

        „ƒƒƒƒƒƒƒƒƒƒƒƒ†

        ‚   STATE    ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     PA     ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚     N      ‚

        ‡ƒƒƒƒƒƒƒƒƒƒƒƒ‰

        ‚        2.00‚

        ŠƒƒƒƒƒƒƒƒƒƒƒƒŒ

The Report Procedure Step’s BY Statement With The NOTSORTED Option
Proc report can produce much of what is done with proc print, proc means, proc sort, proc tabulate, and the data step.

CODE    proc report data=DS1 nowd headline headskip ls=90 ps=50 split='*' center;

        column state  amt;

        by state NOTSORTED; 

        break after state  / skip;

        define state  / 'StatE'   group order=internal width=5;

        define amt    / 'AmounT'   width=10;  run;
OUTPUT  The SAS System                                     1

        20:42 Monday, December 26, 2005   
        ---- STATE=PA ----------------------------------------

        StatE      AmounT

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        PA            300

        The SAS System                                     2

        ---- STATE=CA ----------------------------------------

        StatE      AmounT

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        CA            500

        The SAS System                                     3

        ---- STATE=MA ----------------------------------------

        StatE      AmounT

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        MA            600

        The SAS System                                     4

        ---- STATE=PA ----------------------------------------

        StatE      AmounT

        ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

        PA            400

CODE    proc report data=DS1 nowd headline headskip ls=132 ps=50 split=’*’ center;

        column var1 var2 var3 var4;

        by NOTSORTED; 

        break after var2  / skip;

        break after var1  / skip;

        define var1  / ‘Variable One’   group order=internal format=var1fmt. width=15;

        define var2  / ‘Variable Two’   group format=z3. width=18;

        define var3  / ‘Variable Three’ group order=internal format=$var3fmt. width=15;

        define var4  / ‘Variable Four’  format=8.2 width=8 spacing=7; run;

The Transpose Procedure Step’s BY Statement With The NOTSORTED Option
The variables specified in the BY statement of a proc transpose are used to form BY groups.  Proc transpose does not transpose BY groups.  Instead, for each BY group, proc transpose creates an observation for each variable that it transposes.  If the BY statement’s NOTSORTED option is not specified, the rows must be sorted by all the BY variables or all the BY variables must be indexed appropriately.  The following sample code shows how a ‘wide’ data set can be made ‘long’.  In this example, so as not to loose the link between 200 and 240 (the association between each AMT with a specific WT), a variable (ID) is first added to uniquely identify each row in the original data set. 

DATA SET: DS6     ST   AMT   WT
                  PA   200   240 

                  PA   100   220

                  CA   300   130

                  CA   200   175

                  MA   200   180

                  MA   400   195

                  PA   100   105

                  PA   300   101

CODE    data DS6; set DS6;  

        ID=_N_; run;

        proc transpose data=ds6 out=ds6t (rename=(_name_= attrib ));

        by id st NOTSORTED;

        var amt wt; run;

        proc print data=ds6t noobs; run;

OUTPUT  The SAS System    18:59 Monday, December 26, 2005   1

        ID    ST     attrib    COL1

         1    PA     AMT       200

         1    PA     WT        240

         2    PA     AMT       100

         2    PA     WT        220

         3    CA     AMT       300

         3    CA     WT        130

         4    CA     AMT       200

         4    CA     WT        175

         5    MA     AMT       200

         5    MA     WT        180

         6    MA     AMT       400

         6    MA     WT        195

         7    PA     AMT       100

         7    PA     WT        105

         8    PA     AMT       300

         8    PA     WT        101

The PRINT Procedure Step With The ID Statement , BY Statement, And The NOTSORTED Option
For proc print, when you use an ID statement, the observation numbers are not printed.  The variables in the ID variable list are listed on the left-hand side of the report.  If all BY variables appear in the same order at the beginning of the ID statement, proc print uses a special layout.

CODE    proc print data=DS1;

        BY state NOTSORTED;
        ID state; run;

CODE    proc print data=DS1;

        BY state NOTSORTED;
        ID state NOTSORTED;  run;
CODE    proc print data=DS1;

        BY state NOTSORTED; run;

OUTPUT  The SAS System    18:41 Monday, December 26, 2005  1

        ---- STATE=PA --------------

        Obs    AMT

        1     200

        2     100

        ---- STATE=CA -----

        Obs    AMT

        3     300

        4     200

        ---- STATE=MA -----------

        Obs    AMT

        5     200

        6     400

        ---- STATE=PA -----------------

        Obs    AMT

        7     100

        8     300 */

CODE    proc print data=DS1;

        ID state; run;

OUTPUT  The SAS System     18:41 Monday, December 26, 2005  1

        STATE    AMT

        PA      200

        PA      100

        CA      300

        CA      200

        MA      200

        MA      400

        PA      100

        PA      300 

CODE    proc print data=DS1;

        BY state NOTSORTED;
        ID state; run;

OUTPUT  The SAS System     18:41 Monday, December 26, 2005  1

        STATE   AMT

        PA      200

                100

        CA      300

                200

        MA      200

                400

        PA      100

                300
CODE    proc print data=stamts LABEL UNIFORM  ;

        BY st county NOTSORTED; /*Separate section for each by-group*/

        ID st county;           /*Specifies variables to print instead of obs num*/

        PageBy st;              /*Specifies variables in the BY statement. 

                                  when by value changes a page is ejected*/  

        VAR county st amt;      /*Specifies variables and order. Default is ALL*/

        LABEL st='STate'; 

        LABEL county='CountY'; 

        LABEL amt='Amount';  run;

OUTPUT  The SAS System   20:42 Monday, December 26, 2005   1
        STate      CountY        CountY      STate    Amount

        PA           03            03         PA       300

        The SAS System   20:42 Monday, December 26, 2005   2
        STate      CountY        CountY      STate    Amount

        CA           01            01         CA       500

                                   01         CA       100

        The SAS System   20:42 Monday, December 26, 2005   3
        STate      CountY        CountY      STate    Amount

        MA           01            01         MA       600

        The SAS System   20:42 Monday, December 26, 2005   4
        STate      CountY        CountY      STate    Amount

        PA           03            03         PA       600

        PA           02            02         PA       200   

The COMPARE Procedure Step With The ID Statement, BY Statement, and The NOTSORTED Option
Proc compare compares two data sets: the base data set and the comparison data set.  The procedure determines matching variables and matching observations.  Proc compare produces lengthy output.  Use various options to specify the comparisons to make and the degree of detail in the report.  

The compare procedure has two (2) NOTSORTED options – on both the BY statement and the ID statement.   To use the BY statement with proc compare, both the base and comparison data sets must be sorted on the BY variables.   The nature of the comparison depends on whether all BY variables are in the comparison data set and, if so, whether their attributes match those of the BY variables in the base data set.

The NOTSORTED option on the BY statement specifies that the observations have been grouped in some way, but are not necessarily in sorted order.

The ID statement specifies the variables in the base data set that the procedure uses to match observations.  These variables are the ID variables.  The data set must be sorted by the ID variables or have an appropriate index.  ID variables also identify rows on the printed report and in the output data set.  The NOTSORTED option on the ID statement specifies that the observations have been grouped in some way, but are not necessarily in sorted order.  Use the NOTSORTED option on the ID statement, if you do not need to sort or index the data set by the ID variables.  If the NOTSORTED option is specified, proc compare matches the first row in the base data set with the first row in the comparison data set, the second with the second, and so on.  Also, if the NOTSORTED option is used and the ID values of corresponding rows are not the same, proc compare prints an error message and stops processing. 

If the data sets are not sorted or indexed by the common ID variables and you do not specify the NOTSORTED option, proc compare prints a warning message and continues to process the data sets as if the NOTSORTED option had been specified.

DATA SET: DS1COMP   STATE        AMT    
                    PA           200

                    PA           100

                    CA           300

                    CA           900

                    MA           200

                    MA           400

                    Pa           100

                    PA           300

CODE    proc compare base=ds1 compare=ds1comp;

        by state NOTSORTED;

        run;

OUTPUT  The SAS System     19:27 Monday, December 26, 2005   1

        The COMPARE Procedure

        Comparison of WORK.DS1 with WORK.DS1COMP(Method=EXACT)

        Data Set Summary

        Dataset                Created          Modified  NVar    NObs

        WORK.DS1      26DEC06:18:53:38  26DEC06:18:53:38     2       8

        WORK.DS1COMP  26DEC05:18:53:51  26DEC05:18:53:51     2       8

        Variables Summary

        Number of Variables in Common: 2.

        Number of BY Variables: 1.

        The COMPARE Procedure

        Comparison of WORK.DS1 with WORK.DS1COMP(Method=EXACT)

        ----- STATE=PA ---------------------------------------------------

        Observation Summary

        Observation      Base  Compare

        First Obs           1        1

        Last  Obs           2        2

        Number of Observations in Common: 2.

        Total Number of Observations Read from WORK.DS1: 2.

        Total Number of Observations Read from WORK.DS1COMP: 2.

        Number of Observations with Some Compared Variables Unequal: 0.

        Number of Observations with All Compared Variables Equal: 2.

        NOTE: No unequal values were found. 

        All values compared are exactly equal.

        ------- STATE=CA -------------------------------------------------

        Observation Summary

        Observation      Base  Compare

        First Obs           3        3

        First Unequal       4        4

        Last  Unequal       4        4

        Last  Obs           4        4

        Number of Observations in Common: 2.

        Total Number of Observations Read from WORK.DS1: 2.

        Total Number of Observations Read from WORK.DS1COMP: 2.

        Number of Observations with Some Compared Variables Unequal: 1.

        Number of Observations with All Compared Variables Equal: 1.

        Values Comparison Summary

        Number of Variables Compared with All Observations Equal: 0.

        Number of Variables Compared with Some Observations Unequal: 1.

        Total Number of Values which Compare Unequal: 1.

        Maximum Difference: 700.

        All Variables Compared have Unequal Values

        Variable  Type  Len  Ndif   MaxDif

        AMT       NUM     8     1      700

        Value Comparison Results for Variables

        __________________________________________________________

                   ||       Base    Compare

               Obs ||        AMT        AMT      Diff.     % Diff

         ________  ||  _________  _________  _________  _________

                   ||

                4  ||   200.0000   900.0000   700.0000   350.0000

        __________________________________________________________

        ------ STATE=MA ---------------------------------------------------

        Observation Summary

        Observation      Base  Compare

        First Obs           5        5

        Last  Obs           6        6

        Number of Observations in Common: 2.

        Total Number of Observations Read from WORK.DS1: 2.

        Total Number of Observations Read from WORK.DS1COMP: 2.

        Number of Observations with Some Compared Variables Unequal: 0.

        Number of Observations with All Compared Variables Equal: 2.

        NOTE: No unequal values were found. 

        All values compared are exactly equal.
LOG     proc compare base=ds1 compare=ds1comp;

        by state NOTSORTED;  run;

        ERROR: The by-variable STATE from WORK.DS1 has the value PA, but in 

        WORK.DS1COMP it has the value Pa. This is not permissible with the 

        NOTSORTED option, in which all by-groups must match exactly.

        NOTE: The SAS System stopped processing this step because of errors.

        NOTE: There were 7 observations read from the data set WORK.DS1.

        NOTE: There were 7 observations read from the data set WORK.DS1COMP.

In the output the whole fourth BY group (second ‘PA’ BY group) has not been analyzed, because the ‘Pa’ error stops execution.  Even after fixing the discrepancies identified above, it is advisable to run the compare code again.  
Macro Processing And The BY Statement With The NOTSORTED Option
The following code snippet shows the type of thing that can be done with the NOTSORTED option on the BY statement within a macro program.

CODE    %macro dschk(dsn=_last_, var=, class=, by= ); 

        proc transpose data=&dsn  out=_trans_;

        var &var;  

        by &by NOTSORTED;
        %if %length(&class) > 0 %then %do;

        copy &class; %end; 

        %mend dschk;

CONCLUSION

Intermediate level SAS programmers can benefit by knowing the use of the BY statement’s  NOTSORTED option in both data steps and in various procedures. 

APPENDIX 1:  MORE ABOUT THE NOTSORTED KEYWORD

As an extra, the appendix describes the use of the keyword NOTSORTED with respect to a view, an index, and a user-defined format. 

The NOTSORTED Keyword And Views

When defining a view descriptor, a SORT clause can be used to specify the data order.  A SAS BY statement overrides a SORT clause stored in a view descriptor unless the BY statement includes the NOTSORTED option.  A message is written to the LOG window when the NOTSORTED option causes a SORT clause to be ignored.

The NOTSORTED Keyword And Indexes

BY processing allows processing observations in the order of the values of the BY variables.  Indexing a data file allows the use of a BY statement without sorting the data file.  Specify in the BY statement the variables that are indexed.  When there is a BY statement, SAS looks for an index and if one exists, observations are retrieved from the data file in indexed order.  However, a BY statement will not use an index when 

a) the BY statement includes the DESCENDING or NOTSORTED option. 
b) the index was created with the NOMISS option. or 
c) the data file is physically stored in sorted order based on the variables specified in the BY

    statement.

As an aside, using an index to process a BY statement may not always be more efficient than simply sorting the data file, in particular if the data file has a high blocking factor of observations per page.  Consequently, using an index for a BY statement is usually for convenience and not performance.   

The NOTSORTED Keyword And User-Defined Formats 

When SAS resolves a format, it does a scan of the table of unformatted values and then uses the formatted value.  By default when creating a format, SAS sorts the unformatted values before it builds the table.  To see this, list the format with ‘proc format fmtlib; run;’.   When SAS needs to find an unformatted value it does a binary search.  The performance of the binary search depends on the sorted values.  Of course, the larger the table of values the longer the search will take.  

Proc format supports a NOTSORTED option on the VALUE statement that allows the user, who knows the data well, to list most likely values for a format first and that gives improved performance for long lists of format values.  The NOTSORTED option is common to the VALUE, INVALUE, and PICTURE statements.  This option appears in parentheses after the format or informat name.  For example, if you know that unformatted value ‘4’ occurs far more often than the others and ‘5’ occurs by far the least often then you can use the (NOTSORTED) option.   

CODE    proc format; value lklyfmt (NOTSORTED )

        4=’Most Likely’  7=’Next Most Likely’  5=’Least Likely’; run;

In this case, SAS will search the unformatted values sequentially.   Without (NOTSORTED), SAS will build the lookup table with order 4,5,7.  When searching for 4 it would start in the middle and find 5 and then 4 – two actions.  However, using (NOTSORTED), it would find 4 immediately.  Briefly, the (NOTSORTED) option stores the format with the values in the original order and performs format searches sequentially in that original order.  

Use (NOTSORTED) on the VALUE statement, if you can rank the probabilities of certain ranges occurring and you want to save processing time.  Then create the format or informat using that ranking.  Also use (NOTSORTED) if you want to print a description of the format or informat using the FMTLIB option.  Yet again, use (NOTSORTED) if you want to preserve the order that you define when you use the ORDER=DATA option and the PRELOADFMT option to analyze class variables in PROC MEANS, PROC SUMMARY, or PROC TABULATE.

CODE    proc format; value qrtrfmt (NOTSORTED )

        2=’Quarter 2’  3=’Quarter 3’  1=’Quarter 1’  4=’Quarter 4’; run;

        proc summary print sum mean completetypes;

        class qrtr / preloadfmt order=data;

        var MonTot; run;

When using the CPORT and CIMPORT procedures to transport formats or informats between hosts with different collating sequences, SAS automatically sets the (NOTSORTED) option.

Prior to release SAS9, the MULTILABEL option and the NOTSORTED option could not be specified on the same BY statement.  However, with SAS9, you can format class variables by user-defined formats that have been created with both the MULTILABEL and NOTSORTED options.  By specifying in a CLASS statement, say in a proc means step, the options MLF, PRELOADFMT, and ORDER=DATA, the procedure orders the output according to the label order in the format definition. 

CODE    proc format; value agefmt (NOTSORTED MULTILABEL)

        low - 39=’Below 40 years’ 15 - 29=’15 to 29’ 

        30 - 39 =’30 to 39’ 40 - high=’40 and over’; run;

With appreciation, an example suggested by Jonas Bilenas inspired the following.  Using the proc means CLASS statement with the PRELOADFMT option and ORDER=DATA, we can specify a sort order.  Notice the use of the CLASS statement does not maintain the ‘separate ‘PA’ groupings. 

DATA SET: DSB1      STATE        AMT    
                    PA           200

                    PA           100

                    CA           300

                    MA           400

                    PA           300

 CODE    proc format; 

         value $stfmt (NOTSORTED)  ‘MA’=’MA’  ‘PA’=’PA’  ‘CA’=’CA’; run;

         proc means data=dsb1 mean  maxdec=0;

         class state  / preloadfmt  order=data;

         format state $stfmt.; 

         var amt; run;

OUTPUT  The SAS System      20:42 Monday, December 26, 2005   1

         The MEANS Procedure

         Analysis Variable: AMT

         STATE    N Obs          Mean

         ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

         MA         1             400

         PA         3             200

         CA         1             300

         ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

Introduction 

· We discuss a data analysis problem and present 

   2 solutions that are based on the power of the

   NOTSORTED option of the BY statement.

· Next, we list examples, discussed in detail in the associated paper, of how the NOTSORTED option of the BY statement impacts some of the more common procedure steps.

· Associated with this poster, on the PhilaSUG site, is a 30-page Word paper that contains this poster, the supporting code, and much more.

________________________________________________________________________

________________________________________________________________________

Group Processing

· Variables in a BY statement are called 

   BY variables.

· A BY group is a set of contiguous rows that have the same values for all BY variables.

· The BY variables are used to define BY groups.

______________________________________________________________________

The Data Step With a BY Statement

· The BY statement applies only to the SET, MERGE, UPDATE, or MODIFY statement that immediately precedes it.

· Only one BY statement can accompany each of the SET, MERGE, UPDATE, or MODIFY statements. 

· Without BY’s NOTSORTED option, the data set must be sorted by the BY variables or indexed appropriately. 

________________________________________________________________________

The Data Step With BY’s NOTSORTED

· BY’s NOTSORTED option informs SAS the rows are grouped by the BY variables but are not in ascending or descending sorted order.

· Also, if rows with the same values for the BY variables are not contiguous, a data step with BY-NOTSORTED processing treats each contiguous set as a separate BY group.
________________________________________________________________________

________________________________________________________________________The Data Analysis Problem
· GIVEN:                        CREATE:
   Data Set DS1:             Summary Data Set SUMDS2:

   STATE   AMT              STATE  TOTAMT

   PA          200                PA         300

   PA          100                CA         500

   CA          300                MA        600

   CA          200                PA         400

   MA          200

   MA          400                 # not sorted

   PA           200                 # 4 contiguous by groups

   PA           200             

________________________________________________________________________

A Data Step Solution Using BY’s NOTSORTED

· CODE:

    data SUMDS2 (drop = AMT);

    set DS1;

    by state NOTSORTED;

    if  first.state the TOTAMT = 0;

    totamt + amt;

    if last.state then output;    run;

________________________________________________________________________

________________________________________________________________________

The Procedure Step With a BY Statement

· Only one BY statement can be used in each procedure step.

· A procedure step creates output for each BY group. 

· Without BY’s NOTSORTED option, the data set must be sorted by the BY variables or indexed appropriately. 

________________________________________________________________________

The Procedure Step With BY’s NOTSORTED

· BY’s NOTSORTED option informs SAS that the rows are grouped by the BY variables but are not in ascending or descending sorted order.

· Also, if rows with the same values for the BY variables are not contiguous, a procedure step with BY-NOTSORTED processing treats each contiguous set as a separate BY group.
________________________________________________________________________

________________________________________________________________________

Procedure Steps With BY’s NOTSORTED

· Here are a few procedures supporting BY’s NOTSORTED option:  btl,  calendar,  chart,  compare,  corr,  forms,  freq,  glm,  means,  mi,  nested,  plot,  print,  rank,  report,  score,  standard,  summary,  tabulate,  timeplot,  transpose,  and  univariate. 

· As the usage of the BY statement differs in each procedure, refer to the SAS documentation for details.

________________________________________________________________________

A Proc Means Solution Using BY’s NOTSORTED

· CODE:

   proc means data=DS1  sum  

                        maxdec=0  noprint;

   by state NOTSORTED;

   var amt;

   output out=SUMDS2 

               (drop= _type_  _freq_ )

               sum(amt)=TOTAMT;      run;

________________________________________________________________________

________________________________________________________________________

More Examples and Closing

· The associated paper has many examples of how the NOTSORTED option impacts procedure processing in the more common procedure steps:

   a) sort           procedure             b) print   procedure

   c) univariate procedure             d) freq    procedure

   e) tabulate    procedure             f) report procedure

   g) compare  procedure
· Contacts:    karunasamudral@gmail.com 

   or                 gmgiddings2@yahoo.com 

________________________________________________________________________

